Computation of the constant in the Gagliardo-Nirenberg inequality (d=1)

In[∘]:= `u[x_] := Cosh[x]^(-2/(p-2))`

In[∘]:= $\text{Simplify}\left[\text{FullSimplify}\left[\text{PowerExpand}\left[\frac{-(-2+p)^2 u''[x] + 4 u[x] - 2 p\, u[x]^{p-1}}{u[x]}\right]\right] /. \left\{\text{Sech}[x] \to \frac{1}{C}, \text{Tanh}[x]^2 \to 1 - \frac{1}{C^2}\right\}\right]$

Out[∘]= 0

In[∘]:= $\text{FullSimplify}\left[2\, \text{Integrate}\left[\text{Cosh}[x]^{-q}, \{x, 0, \infty\}, \text{Assumptions} \to q > 0\right]\right]$

Out[∘]= $\frac{\sqrt{\pi}\, \text{Gamma}\left[\frac{q}{2}\right]}{\text{Gamma}\left[\frac{1+q}{2}\right]}$

In[∘]:= $f[q\_] := \frac{\sqrt{\pi}\, \text{Gamma}\left[\frac{q}{2}\right]}{\text{Gamma}\left[\frac{1+q}{2}\right]}$

In[∘]:= $u'[x]^2 /. \text{Sinh}[x]^2 \to \text{Cosh}[x]^2 - 1$

Out[∘]= $\frac{4\, \text{Cosh}[x]^{-2-\frac{4}{-2+p}} \left(-1 + \text{Cosh}[x]^2\right)}{(-2+p)^2}$

In[∘]:= $\text{I1} = \text{FullSimplify}\left[\frac{4\left(-f\left[\frac{4}{-2+p} + 2\right] + f\left[\frac{4}{-2+p}\right]\right)}{(-2+p)^2}\right]$

Out[∘]= $\frac{\sqrt{\pi}\, \text{Gamma}\left[2 + \frac{2}{-2+p}\right]}{p\, \text{Gamma}\left[\frac{3}{2} + \frac{2}{-2+p}\right]}$

In[∘]:= $u[x]^2$

Out[∘]= $\text{Cosh}[x]^{-\frac{4}{-2+p}}$

In[∘]:= $\text{I2} = f\left[\frac{4}{-2+p}\right]$

Out[∘]= $\frac{\sqrt{\pi}\, \text{Gamma}\left[\frac{2}{-2+p}\right]}{\text{Gamma}\left[\frac{1}{2}\left(1 + \frac{4}{-2+p}\right)\right]}$

In[∘]:= $\text{PowerExpand}\left[u[x]^p\right]$

Out[∘]= $\text{Cosh}[x]^{-\frac{2p}{-2+p}}$

In[∘]:= $\text{Ip} = f\left[\frac{2\,p}{-2+p}\right]$

Out[∘]= $$\frac{\sqrt{\pi}\;\text{Gamma}\left[\frac{p}{-2+p}\right]}{\text{Gamma}\left[\frac{1}{2}\left(1+\frac{2\,p}{-2+p}\right)\right]}$$

In[∘]:= $\text{FullSimplify}\left[\text{PowerExpand}\left[\frac{\text{I1}^{\frac{p-2}{2p}}\;\text{I2}^{\frac{p+2}{2p}}}{\text{Ip}^{\frac{2}{p}}}\right]\right];$

$\% \,/.\, \left\{\text{Gamma}\left[\frac{1}{2}+\frac{2}{-2+p}\right]\to X,\ \text{Gamma}\left[\frac{3}{2}+\frac{2}{-2+p}\right]\to\left(\frac{1}{2}+\frac{2}{-2+p}\right)X,\right.$

$\text{Gamma}\left[2+\frac{2}{-2+p}\right]\to\left(1+\frac{2}{-2+p}\right)\frac{2}{-2+p}\,Y,\ \text{Gamma}\left[1+\frac{2}{-2+p}\right]\to\frac{2}{-2+p}\,Y,$

$\left.\text{Gamma}\left[\frac{p}{-2+p}\right]\to\frac{2}{-2+p}\,Y,\ \text{Gamma}\left[\frac{2}{-2+p}\right]\to Y\right\};$

$\text{FullSimplify}\left[\text{PowerExpand}\left[\sqrt{\%}\right]\right];$

$\text{CGNS} = \% \,/.\, \left\{X\to\text{Gamma}\left[\frac{1}{2}+\frac{2}{-2+p}\right],\ Y\to\text{Gamma}\left[\frac{2}{-2+p}\right]\right\}$

Out[∘]= $$2^{\frac{1}{2}-\frac{3}{p}}\,(-2+p)^{-\frac{-6+p}{4p}}\,p^{-\frac{-2+p}{4p}}\left(\frac{p}{-2+p}\right)^{\frac{-2+p}{4p}}\left(\frac{2+p}{-2+p}\right)^{-\frac{-6+p}{4p}}\pi^{\frac{-2+p}{4p}}\,\text{Gamma}\left[\frac{1}{2}+\frac{2}{-2+p}\right]^{-\frac{1}{2}+\frac{1}{p}}\text{Gamma}\left[\frac{2}{-2+p}\right]^{\frac{1}{2}-\frac{1}{p}}$$

In[∘]:= $\text{Simplify}\left[\text{FullSimplify}\left[\text{PowerExpand}\left[\text{CGNS}^{\frac{4p}{p-2}}\right]\right]\,/.\right.$

$\left.\left\{\text{Gamma}\left[\frac{2}{-2+p}\right]\to\frac{-2+p}{2}\,\text{Gamma}\left[\frac{p}{-2+p}\right],\ \text{Gamma}\left[\frac{3}{2}+\frac{2}{-2+p}\right]\to\text{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]\right\}\right]$

Out[∘]= $$\frac{2^{-\frac{2\,(2+p)}{-2+p}}\,(2+p)^{\frac{2+p}{-2+p}}\,\pi\,\text{Gamma}\left[\frac{p}{-2+p}\right]^2}{(-2+p)\,\text{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]^2}$$

In[∘]:=

Taylor expansion in dimension d ≥ 1 and behavior at the bifurcation point

In[∘]:= $\lambda = d + \epsilon^2;$

Basic integral computation

In[∘]:=
```
J2 = d^(2/(p-2)) (1 + a^2 ϵ^2 + b^2 ϵ^4)
K = d^(2/(p-2)) (d a^2 ϵ^2 + 2 (d + 1) b^2 ϵ^4)
Jp = d^(2/(p-2)) (1 + (p - 1) a^2 ϵ^2 + (p - 1) c ϵ^4)
```

Out[∘]= $d^{\frac{2}{-2+p}} \left(1 + a^2 \epsilon^2 + b^2 \epsilon^4\right)$

Out[∘]= $d^{\frac{2}{-2+p}} \left(a^2 d \epsilon^2 + 2 b^2 (1 + d) \epsilon^4\right)$

Out[∘]= $d^{\frac{2}{-2+p}} \left(1 + a^2 (-1 + p) \epsilon^2 + c (-1 + p) \epsilon^4\right)$

In[∘]:=
```
Resc = {c → b^2 + (p - 2) Sqrt[2 d/(3 + d)] a^2 b - (p - 2) (d + p)/(2 (3 + d)) a^4};
FullSimplify[
 Normal[Series[d^(2/(-2+p)) (1 + 1/2 p (p - 1) (a^2 ϵ^2 + b^2 ϵ^4) + 1/6 p (p - 1) (p - 2) 3 Sqrt[2 d/(d + 3)] a^2 b ϵ^4 +
      1/24 p (p - 1) (p - 2) (p - 3) 3 (d + 1)/(d + 3) a^4 ϵ^4)^(2/p) - Jp, {ϵ, 0, 4}]] /. Resc ]
```

Out[∘]= 0

Optimization at $\theta = 0$

In[∘]:=
```
Res = Simplify[Normal[Series[λ - ((p - 2) K + λ J2)/Jp, {ϵ, 0, 4}]] /. Resc];
Simplify[
  Normal[Series[Solve[{D[Res, a] == 0, D[Res, b] == 0}, {a, b}], {ϵ, 0, 1}]]];
Resab = {a → Sqrt[(d + 2) (d + 3)/(d (d + 1) (p - 1) (2 p - d (p - 2)))], b → Sqrt[d (d + 3)]/(Sqrt[2] (d + 1) (2 p - d (p - 2)))}
FullSimplify[PowerExpand[Normal[Series[{D[Res, a], D[Res, b]}, {ϵ, 0, 3}]]]]
FullSimplify[PowerExpand[Res /. Resab]]
```

Out[∘]= $\left\{a \to \sqrt{\frac{(2+d)(3+d)}{d (1+d)(-1+p)(-d(-2+p)+2p)}},\ b \to \frac{\sqrt{d(3+d)}}{\sqrt{2}(1+d)(-d(-2+p)+2p)}\right\}$

Out[∘]= {0, 0}

Out[∘]= $-\frac{(2+d)(3+d)(-2+p)\epsilon^4}{2 d (1+d)(d(-2+p)-2p)(-1+p)}$

Reparametrization

In[∘]:=
```
Nu[x_] := Normal[
    Series[Simplify[Normal[Series[K/J2, {ϵ, 0, 4}]] /. Resab], {ϵ, 0, 4}]] /. ϵ → Sqrt[x]
Nu[
 x]
```

Out[∘]= $-\frac{(2+d)\ (3+d)\ x}{(1+d)\ (-1+p)\ (-2\,d-2\,p+d\,p)}+\frac{(3+d)\ \left(-(2+d)^2\ (3+d)+d^2\ (1+d)\ (-1+p)^2\right)\ x^2}{d\ (1+d)^2\ (d\ (-2+p)-2\,p)^2\ (-1+p)^2}$

In[∘]:=
```
mu = Normal[Series[FullSimplify[PowerExpand[Jp^((p-2)/2)]], {ϵ, 0, 4}]];
Mu[x_] := Simplify[mu] /. ϵ → Sqrt[x]
Mu[x]
Simplify[{Mu[0], Mu'[0], Mu''[0] / 2}]
Simplify[% /. Resc]
```

Out[∘]= $\frac{1}{8}\,d\ \left(8+4\,a^2\ (-2+p)\ (-1+p)\ x+\left(2-3\,p+p^2\right)\ \left(4\,c+a^4\ \left(4-5\,p+p^2\right)\right)\ x^2\right)$

Out[∘]= $\left\{d,\ \frac{1}{2}\,a^2\,d\ (-2+p)\ (-1+p),\ \frac{1}{8}\,d\ \left(2-3\,p+p^2\right)\ \left(4\,c+a^4\ \left(4-5\,p+p^2\right)\right)\right\}$

Out[∘]= $\left\{d,\ \frac{1}{2}\,a^2\,d\ (-2+p)\ (-1+p),\ \frac{1}{8}\,d\ \left(2-3\,p+p^2\right)\ \left(4\,b^2+4\,\sqrt{2}\ a^2\,b\ \sqrt{\frac{d}{3+d}}\ (-2+p)+\frac{a^4\ \left(12-11\,p+p^2+d\ \left(8-7\,p+p^2\right)\right)}{3+d}\right)\right\}$

In[∘]:=
```
Lambda[x_] := θ (d + x) - (1 - θ) (p - 2) Nu[x]
Lambda[x]
```

Out[∘]= $-(-2+p)\ \left(-\frac{(2+d)\ (3+d)\ x}{(1+d)\ (-1+p)\ (-2\,d-2\,p+d\,p)}+\frac{(3+d)\ \left(-(2+d)^2\ (3+d)+d^2\ (1+d)\ (-1+p)^2\right)\ x^2}{d\ (1+d)^2\ (d\ (-2+p)-2\,p)^2\ (-1+p)^2}\right)\ (1-\theta)+(d+x)\ \theta$

In[∘]:=
```
Simplify[θ (d + x) - (1 - θ) (p - 2) Nu[x]]
ResLambda = Simplify[{%, D[%, x], D[%, {x, 2}] / 2} /. x → 0]
```

Out[∘]=
$$-\left(\left((3+d)\,(-2+p)\,x\left(-(1+d)\,(2+d)\,(d\,(-2+p)-2\,p)\,(-1+p)-\frac{(2+d)^2\,(3+d)\,x}{d}+d\,(1+d)\,(-1+p)^2\,x\right)(1-\theta)\right)\Big/\left((1+d)^2\,(d\,(-2+p)-2\,p)^2\,(-1+p)^2\right)\right)+(d+x)\,\theta$$

Out[∘]=
$$\left\{d\,\theta,\ \frac{(2+d)\,(3+d)\,(-2+p)\,(1-\theta)}{(1+d)\,(d\,(-2+p)-2\,p)\,(-1+p)}+\theta,\ -\frac{(3+d)\left(-\frac{(2+d)^2\,(3+d)}{d}+d\,(1+d)\,(-1+p)^2\right)(-2+p)\,(1-\theta)}{(1+d)^2\,(d\,(-2+p)-2\,p)^2\,(-1+p)^2}\right\}$$

In[∘]:=
```
Simplify[Normal[Series[θ Mu[x]^(1/θ) (d + x + (p - 2) Nu[x])^(1 - 1/θ), {x, 0, 2}]] /. Resc];
Simplify[% /. Resab];
ResM = Simplify[{%, D[%, x], D[%, {x, 2}] / 2} /. x → 0]
```

Out[∘]=
$$\Big\{d\,\theta,\ \left(d^2\,(-2+p)\,(5-4\,\theta+p\,(-3+2\,\theta))-2\left(18-12\,\theta+p^2\,(1+2\,\theta)+p\,(-13+4\,\theta)\right)-d\left(34-24\,\theta+p^2\,(3+2\,\theta)+3\,p\,(-9+4\,\theta)\right)\right)\Big/(2\,(1+d)\,(d\,(-2+p)-2\,p)\,(-1+p)),\ \Big(-8\,(2+d)\,(3+d)\,((1+d)\,(d\,(-2+p)-2\,p)\,(1-p)+(2+d)\,(3+d)\,(-2+p))\,(1-p)\,(2-p)\,(1-\theta)+8\,(1-\theta)\left(((1+d)\,(d\,(-2+p)-2\,p)\,(1-p)+(2+d)\,(3+d)\,(-2+p))^2-2\,(3+d)\left((2+d)^2\,(3+d)-d^2\,(1+d)\,(-1+p)^2\right)(2-p)\,\theta\right)+2\,(3+d)\left(2-3\,p+p^2\right)\Big(2\,(2+d)^2\,(3+d)-3\,(2+d)^2\,(3+d)\,p+(2+d)^2\,(3+d)\,p^2+2\,(2+d)^2\,(3+d)\,\theta-2\,(2+d)^2\,(3+d)\,p\,\theta+2\left(2\,d\,(2+d)\,\sqrt{\frac{d}{3+d}}\,\sqrt{d\,(3+d)}\,(-2+p)\,(-1+p)+d^3\,(-1+p)^2-(2+d)^2\,(-2+p)\,(d+p)\right)\theta\Big)\Big)\Big/\left(16\,d\,(1+d)^2\,(d\,(-2+p)-2\,p)^2\,(-1+p)^2\,\theta\right)\Big\}$$

In[∘]:=
```
Simplify[ResLambda[[2]] - ResM[[2]]]
```

Out[∘]=
$$\frac{2\,(-6+p)+3\,d^2\,(-2+p)+d\,(-14+3\,p)}{2\,(1+d)\,(d\,(-2+p)-2\,p)}$$

In[∘]:=
```
M[x_] := ResM
{M[x], M'[x], M''[x] / 2} /. x → 0
```

Out[∘]=
$$\left\{\left\{d\,\theta,\ \left(d^2\,(-2+p)\,(5-4\,\theta+p\,(-3+2\,\theta))-2\,\left(18-12\,\theta+p^2\,(1+2\,\theta)+p\,(-13+4\,\theta)\right)-d\,\left(34-24\,\theta+p^2\,(3+2\,\theta)+3\,p\,(-9+4\,\theta)\right)\right)\Big/\left(2\,(1+d)\,(d\,(-2+p)-2\,p)\,(-1+p)\right),\right.\right.$$
$$\left(-8\,(2+d)\,(3+d)\,((1+d)\,(d\,(-2+p)-2\,p)\,(1-p)+(2+d)\,(3+d)\,(-2+p))\,(1-p)\,(2-p)\,(1-\theta)+8\,(1-\theta)\,\left(((1+d)\,(d\,(-2+p)-2\,p)\,(1-p)+(2+d)\,(3+d)\,(-2+p))^2-2\,(3+d)\,\left((2+d)^2\,(3+d)-d^2\,(1+d)\,(-1+p)^2\right)\,(2-p)\,\theta\right)+2\,(3+d)\,\left(2-3\,p+p^2\right)\left(2\,(2+d)^2\,(3+d)-3\,(2+d)^2\,(3+d)\,p+(2+d)^2\,(3+d)\,p^2+2\,(2+d)^2\,(3+d)\,\theta-2\,(2+d)^2\,(3+d)\,p\,\theta+2\left(2\,d\,(2+d)\,\sqrt{\frac{d}{3+d}}\,\sqrt{d\,(3+d)}\,(-2+p)\,(-1+p)+d^3\,(-1+p)^2-(2+d)^2\,(-2+p)\,(d+p)\right)\theta\right)\right)\Big/$$
$$\left.\left(16\,d\,(1+d)^2\,(d\,(-2+p)-2\,p)^2\,(-1+p)^2\,\theta\right)\right\},\ \{0,0,0\},\ \{0,0,0\}\Big\}$$

Discussion

In[∘]:=
```
Simplify[Lambda'[0]]
{% /. θ → 1, Simplify[Solve[% == 0, θ][[1]]]}
```

Out[∘]=
$$-\frac{(2+d)\,(3+d)\,(-2+p)\,(-1+\theta)}{(1+d)\,(d\,(-2+p)-2\,p)\,(-1+p)}+\theta$$

Out[∘]=
$$\left\{1,\ \left\{\theta\to-\frac{(2+d)\,(3+d)\,(-2+p)}{d^2\,(-2+p)^2-2\,\left(-6+2\,p+p^2\right)-d\,\left(-12+6\,p+p^2\right)}\right\}\right\}$$

In[∘]:=
```
Simplify[M'[0] /. Resab]
{% /. θ → 1, Simplify[Solve[% == 0, θ][[1]]]}
```

Out[∘]= {0, 0, 0}

Out[∘]= {{0, 0, 0}, {}}

In[∘]:=
```
MM = Series[θ d (1 + x/d - γ/d x^2)^(1/θ) (1 + x/d + (p - 2)/d ν1 x + (p - 2)/(2 d) ν2 x^2)^(1-1/θ), {x, 0, 2}]
D[Normal[MM], {x, 2}];
Simplify[{D[%, ν1], D[%, ν2]}];
1/2 %[[1]] ν1 + %[[2]] ν2;
Simplify[% + (-2 + p) (1 - θ) ν2]
```

Out[∘]= $d\,\theta + (\theta + 2\,\nu1 - p\,\nu1 - 2\,\theta\,\nu1 + p\,\theta\,\nu1)\,x + \frac{1}{2\,d\,\theta}\left(-2\,d\,\gamma\,\theta + 4\,\nu1^2 - 4\,p\,\nu1^2 + p^2\,\nu1^2 - 4\,\theta\,\nu1^2 + 4\,p\,\theta\,\nu1^2 - p^2\,\theta\,\nu1^2 + 2\,d\,\theta\,\nu2 - d\,p\,\theta\,\nu2 - 2\,d\,\theta^2\,\nu2 + d\,p\,\theta^2\,\nu2\right)x^2 + O[x]^3$

Out[∘]= $-\frac{(-2+p)^2\,(-1+\theta)\,\nu1^2}{d\,\theta}$

Taylor expansion in dimension d = 1 and behaviour at the bifurcation point

In[∘]:=
```
λ = 1 + ϵ^2;
u[x_] := 1 + Sqrt[2 a] ϵ Cos[x] + Sqrt[2] b ϵ^2 Cos[2 x]
```

In[∘]:=
```
J2 = 1/(2 π) Integrate[u[x]^2, {x, 0, 2 π}]
K = 1/(2 π) Integrate[u'[x]^2, {x, 0, 2 π}]
Normal[Series[u[x]^p, {ϵ, 0, 4}]];
1/(2 π) Integrate[%, {x, 0, 2 π}];
Jp = Simplify[Normal[Series[%^(2/p), {ϵ, 0, 4}]]]
```

Out[∘]= $1 + a\,\epsilon^2 + b^2\,\epsilon^4$

Out[∘]= $\epsilon^2\,(a + 4\,b^2\,\epsilon^2)$

Out[∘]= $1 + a\,(-1+p)\,\epsilon^2 - \frac{1}{8}(-1+p)\left(-8\,b^2 - 4\sqrt{2}\,a\,b\,(-2+p) + a^2\,(-2 - p + p^2)\right)\epsilon^4$

In[∘]:=
```
Res = Simplify[Normal[Series[λ - ((p - 2) K + λ J2)/Jp, {ϵ, 0, 4}]]]
Resab = Solve[{D[Res, a] == 0, D[Res, b] == 0}, {a, b}][[1]]
```

Out[∘]= $-\frac{1}{8}(-2+p)\left(24\,b^2 - 4\,a\left(2 + \sqrt{2}\,b\,(-1+p)\right) + a^2\,(-1+p^2)\right)\epsilon^4$

Out[∘]= $\left\{a \to \frac{6}{-2 + p + p^2},\ b \to \frac{1}{\sqrt{2}\,(2+p)}\right\}$

In[∘]:=
```
Simplify[K/J2 /. Resab];
ν = Simplify[Normal[Series[%, {ϵ, 0, 4}]]]
```

Out[∘]= $$\frac{2\,\epsilon^2\left(-6-17\,\epsilon^2+p\left(3-2\,\epsilon^2\right)+p^2\left(3+\epsilon^2\right)\right)}{\left(-2+p+p^2\right)^2}$$

In[∘]:=
```
θ λ - (1 - θ) (p - 2) ν /. ϵ → Sqrt[x];
Simplify[{% /. x → 0, Solve[D[%, x] == 0, θ][[1]]} /. x → 0]
```

Out[∘]= $$\left\{\theta,\left\{\theta\to\frac{6\,(-2+p)}{-14+7\,p+p^2}\right\}\right\}$$

In[∘]:=
```
Plot[{(p - 2)/(2 p), 6 (-2 + p)/(-14 + 7 p + p^2)}, {p, 2, 10}, PlotStyle → {Automatic, {Black, Dotted}}]
```

Out[∘]= 

In[∘]:=

Figures

The range of the parameters

In[∘]:=
```
Show[Plot[{3 (p - 2)/(4 p - 7), 2}, {p, 2, 15},
  PlotRange → All, Filling → {1 → {{2}, GrayLevel[0.7]}}],
 Plot[{3 (p - 2)/(4 p - 7), (p - 2)/(2 p)}, {p, 2, 15}, PlotStyle → {Black, Black},
  PlotRange → All, Filling → {1 → {{2}, GrayLevel[0.9]}}],
 ListLinePlot[{{0, 1}, {15, 1}}, PlotStyle → {Black, Thin}],
 AxesOrigin → {0, 0}, PlotRange → {{0, 15}, {0, 1.8}}]
```

Out[∘]=



In[•]:=
```
Rng[d_] := Show[Plot[{d (p - 2)/(2 p), 2}, {p, 2, If[d ≤ 2, 15, 2 d/(d - 2)]},
   Filling → {1 → {{2}, GrayLevel[0.9]}}], Plot[{Min[1, 1/(1 + ((d-1)/(d+2))^2 ((p-1) ((2 d^2+1)/(d-1)^2 - p))/(p-2))], 2},
  {p, 2, If[d ≤ 2, 15, 2 d/(d - 2)]}, Filling → {1 → {{2}, GrayLevel[0.7]}}],
  Plot[d (p - 2)/(2 p), {p, 2, If[d ≤ 2, 15, 2 d/(d - 2)]}, PlotStyle → {Black, Thick}], Plot[
   Min[1, 1/(1 + ((d-1)/(d+2))^2 ((p-1) ((2 d^2+1)/(d-1)^2 - p))/(p-2))], {p, 2, If[d ≤ 2, 15, 2 d/(d - 2)]}, PlotStyle → Black],
  ListLinePlot[{{(2 d^2 + 1)/(d - 1)^2, 0}, {(2 d^2 + 1)/(d - 1)^2, 2}}, PlotStyle → {Black, Thin}],
  If[d ≥ 3, ListLinePlot[{{2 d/(d - 2), 0}, {2 d/(d - 2), 2}}, PlotStyle → {Black, Thin}], {}],
  If[d ≥ 3, ListLinePlot[{{2 d/(d - 2), 1}, {2 d/(d - 2), 2}}, PlotStyle → Black], {}],
  ListLinePlot[{{0, 1}, {15, 1}}, PlotStyle → {Black, Thin}],
  AxesOrigin → {0, 0}, PlotRange → {{0, 10}, {0, 1.8}}]
```

In[•]:=
```
Rng[2]
```

Out[•]=



In[∘]:=
```
Show[Rng[3], PlotRange → {{0, 8}, {0, 1.8}}]
```

Out[∘]=


Plotting the branches of solutions in dimension d = 1

In[∘]:=
$$ac[p\_] := \left(\frac{p}{2}\right)^{\frac{1}{p-2}}$$

$$En[a\_, p\_] := \frac{a^p}{p} - \frac{a^2}{2}$$

```
b[a_, p_] := b /. FindRoot[En[b, p] - En[a, p], {b, 1, ac[p]}]
```

In[∘]:=
$$df[a\_, f\_, p\_] := \sqrt{2\,(En[a, p] - En[f, p])}$$

F[a_, p_] :=

$$NIntegrate\left[\left\{\frac{2}{df[a, f, p]}, df[a, f, p], \frac{f^2}{df[a, f, p]}, \frac{f^p}{df[a, f, p]}\right\}, \{f, a, b[a, p]\}\right]$$

In[∘]:=
$$Q[a\_, p\_, \theta\_] := Module\left[\{M = F[a, p]\}, \left\{(p-2)\left(\theta - (1-\theta)\frac{M[[2]]}{M[[3]]}\right)\left(\frac{M[[1]]}{2\pi}\right)^2,\right.\right.$$
$$\left.\left.\left(\frac{M[[1]]}{2}\right)^{-\frac{p-2}{p\theta}}\left((p-2)\left(\frac{M[[1]]}{2\pi}\right)^2\theta\left(1+\frac{M[[2]]}{M[[3]]}\right)\right)\frac{M[[3]]^{\frac{1}{\theta}}}{M[[4]]^{\frac{2}{p\theta}}}\right\}\right]$$

In[∘]:=
```
FQ[p_, θ_, λmax_] := Show[Show[ParametricPlot[Q[a, p, θ], {a, 0.001, 0.9999},
     PlotStyle → Thickness[0.01], PlotRange → All, AspectRatio → 1],
    Plot[λ, {λ, 0, λmax}, PlotStyle → {Brown, Thickness[0.01]}],
    ListLinePlot[{{0, θ}, {1.2 θ, θ}}, PlotStyle → Black],
    ListLinePlot[{{θ, 0}, {θ, 1.2 θ}}, PlotStyle → Black],
    AxesOrigin → {0, 0}], PlotRange → {{0, λmax}, {0, λmax}}]
```

In[∘]:= `FQ[5, 2, 9]`

Out[∘]=



In[∘]:= `FQ[5, 0.8, 6]`

Out[∘]=



In[∘]:= `FQ[5, 0.32, 0.5]`

Out[∘]=



In[∘]:= `FQ[5, 0.3, 0.35]`

Out[∘]=



Taylor expansion in dimension d = 1 and behaviour at the bifurcation point

In[∘]:= $\lambda = 1 + \epsilon^2;$

$u[x\_] := 1 + \sqrt{2\,a}\ \epsilon\, \text{Cos}[x] + \sqrt{2}\ b\, \epsilon^2\, \text{Cos}[2\,x]$

In[∘]:=
```
J2 = 1/(2 π) Integrate[u[x]^2, {x, 0, 2 π}]
K = 1/(2 π) Integrate[u'[x]^2, {x, 0, 2 π}]
Normal[Series[u[x]^p, {ϵ, 0, 4}]];
1/(2 π) Integrate[%, {x, 0, 2 π}];
Jp = Simplify[Normal[Series[%^(2/p), {ϵ, 0, 4}]]]
```

Out[∘]= $1+a\,\epsilon^2+b^2\,\epsilon^4$

Out[∘]= $\epsilon^2\left(a+4\,b^2\,\epsilon^2\right)$

Out[∘]= $1+a\,(-1+p)\,\epsilon^2-\frac{1}{8}\,(-1+p)\left(-8\,b^2-4\,\sqrt{2}\,a\,b\,(-2+p)+a^2\left(-2-p+p^2\right)\right)\epsilon^4$

In[∘]:=
```
Res = Simplify[Normal[Series[λ - ((p - 2) K + λ J2)/Jp, {ϵ, 0, 4}]]]
Resab = Solve[{D[Res, a] == 0, D[Res, b] == 0}, {a, b}][[1]]
```

Out[∘]= $-\frac{1}{8}\,(-2+p)\left(24\,b^2-4\,a\left(2+\sqrt{2}\,b\,(-1+p)\right)+a^2\left(-1+p^2\right)\right)\epsilon^4$

Out[∘]= $\left\{a\to\frac{6}{-2+p+p^2},\ b\to\frac{1}{\sqrt{2}\,(2+p)}\right\}$

In[∘]:=
```
Simplify[Res /. Resab]
μ = λ - %
```

Out[∘]= $\frac{3\,(-2+p)\,\epsilon^4}{-2+p+p^2}$

Out[∘]= $1+\epsilon^2-\frac{3\,(-2+p)\,\epsilon^4}{-2+p+p^2}$

In[∘]:=
```
Simplify[K/J2 /. Resab];
ν = Simplify[Normal[Series[%, {ϵ, 0, 4}]]]
```

Out[∘]= $\frac{2\,\epsilon^2\left(-6-17\,\epsilon^2+p\left(3-2\,\epsilon^2\right)+p^2\left(3+\epsilon^2\right)\right)}{\left(-2+p+p^2\right)^2}$

In[•]:=
```
Lambda = θ λ - (1 - θ) (p - 2) ν;
Simplify[{% /. ϵ → 0, AA = 1/2 D[%, {ϵ, 2}] /. ϵ → 0, BB = 1/24 D[%, {ϵ, 4}] /. ϵ → 0}]
Mu = Simplify[Normal[Series[θ ( (p - 2) ν + λ)^(1-1/θ) μ^(1/θ), {ϵ, 0, 4}]]];
Simplify[{% /. ϵ → 0, CC = 1/2 D[%, {ϵ, 2}] /. ϵ → 0, DD = 1/24 D[%, {ϵ, 4}] /. ϵ → 0}]
Simplify[AA == CC]
```

Out[•]= $\left\{\theta, \frac{12 - 14\,\theta + p^2\,\theta + p\,(-6 + 7\,\theta)}{-2 + p + p^2}, \frac{2\,(-2 + p)\,\left(-17 - 2\,p + p^2\right)\,(-1 + \theta)}{\left(-2 + p + p^2\right)^2}\right\}$

Out[•]= $\left\{\theta, \frac{12 - 14\,\theta + p^2\,\theta + p\,(-6 + 7\,\theta)}{-2 + p + p^2}, \frac{(-2 + p)\,\left(-36 + 76\,\theta - 34\,\theta^2 + p^2\,\theta\,(-5 + 2\,\theta) + p\,\left(18 - 17\,\theta - 4\,\theta^2\right)\right)}{\left(-2 + p + p^2\right)^2\,\theta}\right\}$

Out[•]= True

In[•]:=
```
Simplify[Solve[AA == 0, θ][[1]]]
Solve[(p - 2)/(2 p) == θ /. %, p]
```

Out[•]= $\left\{\theta \to \frac{6\,(-2 + p)}{-14 + 7\,p + p^2}\right\}$

Out[•]= $\{\{p \to -2\}, \{p \to 2\}, \{p \to 7\}\}$

In[•]:=
```
Plot[{(p - 2)/(2 p), (6 (-2 + p))/(-14 + 7 p + p^2)}, {p, 2, 10}, PlotStyle → {Automatic, {Red, Dotted}}]
```

Out[•]=


In[∘]:=
$$\text{P0black} = \text{Show}\left[\text{Plot}\left[\frac{p-2}{2\,p}, \{p, 2, 12\}\right],\ \text{Plot}\left[\frac{6\,(-2+p)}{-14+7\,p+p^2}, \{p, 2, 12\}, \text{PlotStyle} \to \{\text{Automatic}, \{\text{Black}, \text{Dotted}\}\}\right]\right]$$

Out[∘]=



In[∘]:=
$$\theta s = \frac{p-2}{2\,p};$$

$$\text{P0} = \text{Show}\left[\text{Plot}\left[\left\{\frac{p-2}{2\,p},\ 3\,\frac{p-2}{4\,p-7},\ \frac{p-2}{4\,\pi^2}\left(\frac{p^{2/p}}{2}\,(1-\theta s)^{1-\theta s}\,\theta s^{-\theta s}\left(\frac{\sqrt{\pi}\ \text{Gamma}\left[\frac{p}{-2+p}\right]}{\text{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]}\right)^{\frac{p-2}{p}}\right)^{\frac{2\,p}{p-2}},\right.\right.$$
$$\left.\left.(p-2)\,\frac{\left(\frac{p^{\frac{1}{p}}\,\pi^{\frac{1}{2}-\frac{1}{p}}}{\sqrt{2}}\right)^{\frac{1}{\theta s}}}{4\,\pi^2}\right\}, \{p, 2, 12\}\right], \text{P0black}\right]$$

Out[∘]=



Double turning points

In[∘]:=
```
FQdet[p_, θ_, λmax_, r_] := Show[Show[ParametricPlot[Q[a, p, θ], {a, 0.001, 0.9999},
    PlotStyle → Thickness[r], PlotRange → All, AspectRatio → 1],
   Plot[λ, {λ, 0, λmax}, PlotStyle → {Brown, Thickness[r]}],
   ListLinePlot[{{0, θ}, {1.2 θ, θ}}, PlotStyle → {Black, Thickness[r / 2]}],
   ListLinePlot[{{θ, 0}, {θ, 1.2 θ}}, PlotStyle → {Black, Thickness[r / 2]}],
   AxesOrigin → {0, 0}], PlotRange → {{0, λmax}, {0, λmax}}]
```

In[•]:=
```
FQdet[9, 7/18 + 0.01, 0.6, 0.01]
```

Out[•]=



In[•]:=
```
Show[FQdet[9, 7/18 + 0.01, 0.6, 0.005],
 PlotRange → {{0.25, 0.45}, {0.38, 0.45}}, AspectRatio → 7/20]
```

Out[•]=



In[•]:=
```
FF[p_, θ_, δ_] := Show[FQ[p, θ, 1], PlotRange → {{θ - δ, θ + δ}, {θ - δ, θ + δ}}]
```

In[∘]:= FF$\left[9, \frac{7}{18}, 0.5\right]$

Out[∘]=



Gagliardo-Nirenberg constant in dimension d = 1

In[∘]:= $\theta s = \frac{p-2}{2\,p}$;

```
FullSimplify[Integrate[Cosh[x]^-q, {x, -∞, ∞}, Assumptions → q > 0]]
```

g[q_] := $\frac{\sqrt{\pi}\ \text{Gamma}\left[\frac{q}{2}\right]}{\text{Gamma}\left[\frac{1+q}{2}\right]}$

Out[∘]= $\frac{\sqrt{\pi}\ \text{Gamma}\left[\frac{q}{2}\right]}{\text{Gamma}\left[\frac{1+q}{2}\right]}$

In[•]:=
```
f[x_] := Cosh[x]^(-2/(p-2))
FullSimplify[PowerExpand[{f'[x]^2, f[x]^2, f[x]^p}]] /. Sinh[x]^2 → Cosh[x]^2 - 1
FullSimplify[PowerExpand[{(4 (-g[(2 p)/(-2+p)] + g[(2 p)/(-2+p) - 2]))/(-2 + p)^2, g[4/(-2 + p)], g[(2 p)/(-2 + p)]}]];
Res = Simplify[
  % /. {Gamma[1 + p/(-2 + p)] → p/(-2 + p) X, Gamma[3/2 + 2/(-2 + p)] → Y, Gamma[2/(-2 + p)] → (p - 2)/2 X,
    Gamma[1/2 + 2/(-2 + p)] → Y/(1/2 + 2/(-2+p)), Gamma[p/(-2 + p)] → X, Gamma[1/2 + p/(-2 + p)] → Y}]
```

Out[•]=
$$\left\{\frac{4\,\mathrm{Cosh}[x]^{-\frac{2p}{-2+p}}\left(-1+\mathrm{Cosh}[x]^2\right)}{(-2+p)^2},\ \mathrm{Cosh}[x]^{-\frac{4}{-2+p}},\ \mathrm{Cosh}[x]^{-\frac{2p}{-2+p}}\right\}$$

Out[•]=
$$\left\{\frac{\sqrt{\pi}\,X}{(-2+p)\,Y},\ \frac{(2+p)\sqrt{\pi}\,X}{4\,Y},\ \frac{\sqrt{\pi}\,X}{Y}\right\}$$

In[•]:=
```
GN[p_] := FullSimplify[PowerExpand[(Res[[1]]^θs Res[[2]]^(1-θs))/Res[[3]]^(2/p)]] /.
  {X → Gamma[p/(-2 + p)], Y → Gamma[1/2 + p/(-2 + p)]}
Plot[GN[p], {p, 2, 10}, PlotRange → All]
Limit[GN[p], p → 2]
```

Out[•]=


Out[•]= 1

The Gagliardo-Nirenberg constant as a lower estimate for the constant in dimension d = 1

In[∘]:=
$$H[p\_] := \frac{2^{-\frac{4p}{-2+p}} (2+p)^{\frac{2+p}{-2+p}} \text{Gamma}\left[\frac{p}{-2+p}\right]^2}{\pi\, \text{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]^2}$$

CritPlot[p_, Rng_] := Show[
ListLinePlot[{{0, H[p]}, {Rng, H[p]}}, PlotStyle → {Dashed, Thickness[0.007]}],
$FQ\left[p, \frac{p-2}{2p}, Rng\right]$, PlotRange → {{0, Rng}, {0, Rng}}]

In[∘]:= CritPlot[5, 0.4]

Out[∘]=



In[∘]:= CritPlot[5, 0.4]

Out[∘]=



In[∘]:= $N\left[\frac{p-2}{2p} \;/.\; p \to 9\right]$

Out[∘]= 0.388889

In[•]:=
```
Show[CritPlot[9, 0.45], PlotRange → {All, {0.36, 0.43}}, AxesOrigin → {0, 0.36}]
```

Out[•]=



In[•]:=
```
Show[CritPlot[9.7, 0.47], PlotRange → {All, {0.38, 0.45}}, AxesOrigin → {0, 0.38}]
```

Out[•]=



In[•]:=
```
Show[CritPlot[10.1, 0.47], PlotRange → {All, {0.39, 0.46}}, AxesOrigin → {0, 0.39}]
```

Out[•]=



In[•]:=
```
Off[NIntegrate::ncvb]
```

In[•]:= Show[CritPlot[10.8, 0.49], PlotRange → {All, {0.4, 0.49}}, AxesOrigin → {0, 0.4}]

Out[•]=



Gaussian approximation as an estimate of the Gagliardo-Nirenberg constant in dimension d = 1

In[•]:= Off[Solve::ifun]

$$g[x\_] := \frac{e^{-\frac{x^2}{4}}}{(2\pi)^{\frac{1}{4}}}$$

I2g = Integrate[g[x]^2, {x, -∞, ∞}];

Kg = Integrate[g'[x]^2, {x, -∞, ∞}];

Ipg = Integrate[g[x]^p, {x, -∞, ∞}, Assumptions → p > 2];

$$\text{FullSimplify}\left[\text{PowerExpand}\left[\frac{Kg^{\frac{p-2}{2p}}\, I2g^{\frac{p+2}{2p}}}{Ipg^{\frac{2}{p}}}\right]\right]$$

Out[•]=

$$\frac{p^{\frac{1}{p}}\, \pi^{\frac{1}{2}-\frac{1}{p}}}{\sqrt{2}}$$

In[∘]:= $$\text{P3} = \text{Show}\Big[\text{Plot}\Big[\Big\{\frac{p-2}{2p},\ 3\,\frac{p-2}{4p-7},\ \frac{p-2}{4\pi^2}\left(\frac{p^{2/p}}{2}(1-\theta s)^{1-\theta s}\,\theta s^{-\theta s}\left(\frac{\sqrt{\pi}\ \text{Gamma}\left[\frac{p}{-2+p}\right]}{\text{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]}\right)^{\frac{p-2}{p}}\right)^{\frac{2p}{p-2}},\ (p-2)\,\frac{\left(\frac{p^{\frac{1}{p}}\,\pi^{\frac{1}{2}-\frac{1}{p}}}{\sqrt{2}}\right)^{\frac{1}{\theta s}}}{4\pi^2}\Big\},$$

$$\{p, 2, 25\}, \text{PlotRange} \to \{\text{All}, \{0, 0.8\}\}, \text{AxesOrigin} \to \{0, 0\}\Big],$$

$$\text{Plot}\left[\frac{p-2}{2p}, \{p, 2, 9.911091772894673`\}, \text{PlotStyle} \to \text{Thickness}[0.01]\right]\Big]$$

$$\theta\text{GN} = \text{FullSimplify}\left[\frac{p-2}{4\pi^2}\left(\frac{p^{2/p}}{2}(1-\theta s)^{1-\theta s}\,\theta s^{-\theta s}\left(\frac{\sqrt{\pi}\ \text{Gamma}\left[\frac{p}{-2+p}\right]}{\text{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]}\right)^{\frac{p-2}{p}}\right)^{\frac{2p}{p-2}}\right];$$

$$\left\{\text{FindRoot}\left[\frac{p-2}{2p}-\theta\text{GN}, \{p, 10\}\right], \text{FindRoot}\left[3\,\frac{p-2}{4p-7}-\theta\text{GN}, \{p, 10\}\right]\right\}$$

$$\text{FullSimplify}\left[\text{PowerExpand}\left[\frac{p^{\frac{1}{p}}\,\pi^{\frac{1}{2}-\frac{1}{p}}}{\sqrt{2}\left(\frac{2\pi^2}{p}\right)^{\theta s}}\right]\right]$$

```
Solve[% == 1, p][[1]]
N[p /. %]
```

Out[∘]= 

Out[∘]= {{p → 9.91109}, {p → 20.8234}}

Out[∘]= $2^{-1+\frac{1}{p}}\sqrt{p}\ \pi^{-\frac{1}{2}+\frac{1}{p}}$

Out[∘]= $\left\{p \to -\frac{2\,(\text{Log}[2]+\text{Log}[\pi])}{\text{ProductLog}\left[\frac{-\text{Log}[2]-\text{Log}[\pi]}{2\pi}\right]}\right\}$

Out[∘]= 7.8834

In[•]:=
```
P1 = Plot[(p - 2)/(2 p), {p, 2, 9.911091772894673`}, PlotStyle → Thickness[0.01]];
P2 = Show[Plot[{(3 (p - 2))/(4 p - 7), 2}, {p, 2, 12}, PlotRange → {{0, 12}, {0, 0.8}},
    Filling → {1 → {{2}, GrayLevel[0.7]}}], Plot[{(p - 2)/(2 p), (3 (p - 2))/(4 p - 7)}, {p, 2, 12},
    PlotStyle → Thin, PlotRange → All, Filling → {1 → {{2}, GrayLevel[0.9]}}]];
Show[
 P2,
 P0,
 P1]
```

Out[•]=



Plot of the value of θ for which the intersection of the branch with the straight line takes place precisely at θ

```
Off[FindRoot::nlnum]
Off[ReplaceAll::reps]
Off[NIntegrate::nlim]
Off[NIntegrate::ncvb]
Off[FindRoot::lstol]
Off[Power::infy]
Off[Infinity::indet]
Off[NIntegrate::zeroregion]
Off[FindRoot::brmp]
```

In[•]:=
```
ac[p_] := (p/2)^(1/(p-2))
En[a_, p_] := a^p/p - a^2/2
b[a_, p_] := b /. FindRoot[En[b, p] - En[a, p], {b, 1, ac[p]}]
```

```
In[∘]:= df[a_, f_, p_] := Sqrt[2 (En[a, p] - En[f, p])]
        F[a_, p_] :=
         NIntegrate[{2/df[a, f, p], df[a, f, p], f^2/df[a, f, p], f^p/df[a, f, p]}, {f, a, b[a, p]}]
```

```
In[∘]:= Qbar[a_, p_] := Module[{M = F[a, p]}, {(p - 2) (M[[1]]/(2 π))^2,
          (M[[1]]/2)^(-(p-2)/p) ((p - 2) (M[[1]]/(2 π))^2 (1 + M[[2]]/M[[3]])) M[[3]]/M[[4]]^(2/p), (M[[1]]/(2 π))^2 M[[2]]/M[[3]]}]
```

```
In[∘]:= PP[p_] := Module[{M = Qbar[a, p]}, Plot[(M[[1]] - 1) Log[(p - 2) M[[3]] + M[[1]]] -
           ((p - 2) M[[3]] + M[[1]] - 1) Log[M[[2]]], {a, 0.001, 0.99}, PlotRange → All]]
```

```
In[∘]:= PP[6.5]
```

Out[∘]=




```
In[∘]:= PP[8.5]
```

Out[∘]=




```
In[∘]:= RE[p_] :=
         a /. Module[{M = Qbar[a, p]}, FindRoot[(M[[1]] - 1) Log[(p - 2) M[[3]] + M[[1]]] -
             ((p - 2) M[[3]] + M[[1]] - 1) Log[M[[2]]], {a, 0.01, 0.4}]]
```

```
In[•]:= Table[{p, RE[p]}, {p, 4, 8, 0.1}];
        Tbl = Re[Chop[%, 10^-5]]

Out[•]= {{4., 0.945851}, {4.1, 0.927616}, {4.2, 0.928168}, {4.3, 0.928653}, {4.4, 0.94799},
         {4.5, 0.933735}, {4.6, 0.916816}, {4.7, 0.926884}, {4.8, 0.932829},
         {4.9, 0.928811}, {5., 0.925379}, {5.1, 0.919337}, {5.2, 0.928705},
         {5.3, 0.837986}, {5.4, 0.750933}, {5.5, 0.565825}, {5.6, 0.456317},
         {5.7, 0.380021}, {5.8, 0.322885}, {5.9, 0.27814}, {6., 0.242015},
         {6.1, 0.212136}, {6.2, 0.186967}, {6.3, 0.165459}, {6.4, 0.146855},
         {6.5, 0.130602}, {6.6, 0.116285}, {6.7, 0.103588}, {6.8, 0.0922627},
         {6.9, 0.0824355}, {7., 0.0729827}, {7.1, 0.064743}, {7.2, 0.0572896},
         {7.3, 0.0505359}, {7.4, 0.0444097}, {7.5, 0.0388502}, {7.6, 0.0338061},
         {7.7, 0.0292341}, {7.8, 0.0250968}, {7.9, 0.0213627}, {8., 0.0180043}}

In[•]:= ListPlot[Tbl]
```

Out[•]=




```
In[•]:= Table[{Tbl[[k]][[1]], Chop[Qbar[Tbl[[k]][[2]], Tbl[[k]][[1]]], 10^-5]},
          {k, 1, Length[Tbl]}];
        Tbl2 = Table[{Tbl[[k]][[1]],
           Chop[(Tbl[[k]][[1]] - 2) %[[k]][[2]][[3]] /
             ((Tbl[[k]][[1]] - 2) %[[k]][[2]][[3]] + %[[k]][[2]][[1]] - 1), 10^-5]}, {k,
           1, Length[Tbl]}]

Out[•]= {{4., 0.399721}, {4.1, 0.399552}, {4.2, 0.399219}, {4.3, 0.398694},
         {4.4, 0.398045}, {4.5, 0.397317}, {4.6, 0.395955}, {4.7, 0.394999},
         {4.8, 0.393933}, {4.9, 0.39256}, {5., 0.391167}, {5.1, 0.389733}, {5.2, 0.388311},
         {5.3, 0.386669}, {5.4, 0.385263}, {5.5, 0.384137}, {5.6, 0.383368},
         {5.7, 0.382864}, {5.8, 0.382562}, {5.9, 0.382416}, {6., 0.38239},
         {6.1, 0.382459}, {6.2, 0.382604}, {6.3, 0.382809}, {6.4, 0.383063},
         {6.5, 0.383355}, {6.6, 0.383679}, {6.7, 0.384028}, {6.8, 0.384397},
         {6.9, 0.38482}, {7., 0.385182}, {7.1, 0.385591}, {7.2, 0.386009},
         {7.3, 0.386433}, {7.4, 0.386863}, {7.5, 0.387297}, {7.6, 0.387735},
         {7.7, 0.388175}, {7.8, 0.388618}, {7.9, 0.389062}, {8., 0.389509}}
```

In[•]:= `ListLinePlot[Tbl2, PlotStyle → {Brown, Thick}]`

Out[•]=



In[•]:= `Show[P2, P0, P1, ListLinePlot[Tbl2, PlotStyle → {Brown, Thick}]]`

Out[•]=



For which values of *p* is $\kappa(p,\theta_*)$ less than θ_* ?

In[∘]:=
```
Show[Plot[{(p - 2)/(2 p), (p - 2) GN[p]^(1/θs)/(4 π^2)}, {p, 2, 30}, AxesOrigin → {0, 0}],
  ListLinePlot[{{2, 0}, {2, 1}}, PlotStyle → {Black, Thin}],
  PlotRange → {All, {0, 0.45}}]
Show[%, PlotRange → {{2, 10}, {0, 0.3}}]
pcrit = p /. FindRoot[(p - 2)/(2 p) - (p - 2) GN[p]^(1/θs)/(4 π^2), {p, 3, 15}]
```

Out[∘]=


Out[∘]=


Out[∘]= 9.91109

In[∘]:= Plot[

$$\left\{\frac{p-2}{2\,p},\ 3\,\frac{p-2}{4\,p-7},\ \frac{p-2}{4\,\pi^2}\left(\frac{p^{2/p}}{2}\,(1-\theta s)^{1-\theta s}\,\theta s^{-\theta s}\left(\frac{\sqrt{\pi}\ \mathrm{Gamma}\left[\frac{p}{-2+p}\right]}{\mathrm{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]}\right)^{\frac{p-2}{p}}\right)^{\frac{2\,p}{p-2}}\right\},\ \{p,\ 2,\ 25\}\Big]$$

Out[∘]=



In[∘]:= $$\mathrm{FullSimplify}\left[\frac{p-2}{2\,p}-\frac{p-2}{4\,\pi^2}\left(\frac{p^{2/p}}{2}\,(1-\theta s)^{1-\theta s}\,\theta s^{-\theta s}\left(\frac{\sqrt{\pi}\ \mathrm{Gamma}\left[\frac{p}{-2+p}\right]}{\mathrm{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]}\right)^{\frac{p-2}{p}}\right)^{\frac{2\,p}{p-2}}\right];$$

```
FindRoot[%, {p, 10}]
```

$$\mathrm{FullSimplify}\left[3\,\frac{p-2}{4\,p-7}-\frac{p-2}{4\,\pi^2}\left(\frac{p^{2/p}}{2}\,(1-\theta s)^{1-\theta s}\,\theta s^{-\theta s}\left(\frac{\sqrt{\pi}\ \mathrm{Gamma}\left[\frac{p}{-2+p}\right]}{\mathrm{Gamma}\left[\frac{1}{2}+\frac{p}{-2+p}\right]}\right)^{\frac{p-2}{p}}\right)^{\frac{2\,p}{p-2}}\right];$$

```
FindRoot[%, {p, 10}]
```

Out[∘]= {p → 9.91109}

Out[∘]= {p → 20.8234}

Gaussian approximation as an estimate of the Gagliardo-Nirenberg constant in dimension d ≥ 1

In[•]:= Off[Solve::ifun]

$$\text{Sp}[d\_] := \frac{2\,\pi^{\frac{d}{2}}}{\text{Gamma}\left[\frac{d}{2}\right]};$$

$$\theta s = d\,\frac{p-2}{2\,p};$$

$$g[x\_] := \frac{e^{-\frac{x^2}{4}}}{(2\,\pi)^{\frac{1}{4}}}$$

$$\text{I2g} = \text{Integrate}\left[x^{d-1}\, g[x]^2,\ \{x, 0, \infty\},\ \text{Assumptions} \to d \geq 1\right];$$

$$\text{Kg} = \text{Integrate}\left[\frac{1}{4}\, x^{d+1}\, g[x]^2,\ \{x, 0, \infty\},\ \text{Assumptions} \to d \geq 1\right];$$

$$\text{Ipg} = \text{Integrate}\left[x^{d-1}\, g[x]^p,\ \{x, 0, \infty\},\ \text{Assumptions} \to p > 2\ \&\&\ d \geq 1\right];$$

CGNS =

$$\text{FullSimplify}\left[\text{PowerExpand}\left[\left(\frac{\text{Sp}[d]}{\text{Sp}[d+1]}\right)^{1-\frac{2}{p}} \frac{\text{Kg}^{d\,\frac{p-2}{2\,p}}\ \text{I2g}^{\frac{2\,d-p\,(d-2)}{2\,p}}}{\text{Ipg}^{\frac{2}{p}}}\right],\ \text{Assumptions} \to d \geq 1\right]$$

Out[•]= $$2^{-1-\frac{d}{2}+\frac{2}{p}}\, d^{\frac{d\,(-2+p)}{2\,p}}\, p^{d/p}\, \pi^{-\frac{1}{2}+\frac{1}{p}}\, \text{Gamma}\left[\frac{1+d}{2}\right]^{\frac{-2+p}{p}}$$

In[•]:=

In[•]:= $$\text{FullSimplify}\left[\text{PowerExpand}\left[(p-2)\ \text{CGNS}^{\frac{1}{\theta s}}\right]\right]$$

$$\text{Limit}\left[\text{CGNS}^{\frac{1}{\theta s}},\ p \to 2\right]$$

$$\text{Plot}\left[\left\{1,\ 2^{-\frac{2+d}{d}}\ \frac{4}{d}\ e\ \pi^{-1/d}\ \text{Gamma}\left[\frac{1+d}{2}\right]^{2/d}\right\},\ \{d, 1, 10\}\right]$$

Out[•]= $$2^{\frac{4-(2+d)\,p}{d\,(-2+p)}}\ d\ (-2+p)\ p^{\frac{2}{-2+p}}\ \pi^{-1/d}\ \text{Gamma}\left[\frac{1+d}{2}\right]^{2/d}$$

Out[•]= $$2^{-\frac{2+d}{d}}\ d\ e\ \pi^{-1/d}\ \text{Gamma}\left[\frac{1+d}{2}\right]^{2/d}$$

Out[•]=



In[ ]:= GP[d_] :=

$$\text{Plot}\left[\left\{\frac{1}{d}\,2^{-1-\frac{d}{2}+\frac{2}{p}}\,d^{\frac{d\,(-2+p)}{2\,p}}\,p^{d/p}\,\pi^{-\frac{1}{2}+\frac{1}{p}}\,\text{Gamma}\left[\frac{1+d}{2}\right]^{\frac{-2+p}{p}},\ d\,\frac{p-2}{2\,p}\right\},\ \left\{p,\ 2,\ \text{If}\left[d>2,\ \frac{2\,d}{d-2},\ 10\right]\right\}\right]$$

In[ ]:= {GP[1], GP[2], GP[3]}

Out[ ]=



In[ ]:= Show[Plot[

$$\left\{2,\ p\ /.\ \text{FindRoot}\left[\frac{1}{d}\,2^{-1-\frac{d}{2}+\frac{2}{p}}\,d^{\frac{d\,(-2+p)}{2\,p}}\,p^{d/p}\,\pi^{-\frac{1}{2}+\frac{1}{p}}\,\text{Gamma}\left[\frac{1+d}{2}\right]^{\frac{-2+p}{p}}-d\,\frac{p-2}{2\,p},\ \{p,\ 2\}\right][[1]]\right\},$$

{d, 1, 10}, PlotStyle → Thickness[0.01], PlotRange → All, AxesOrigin → {1, 0}]]

Out[ ]=



In[ ]:=

$$\text{FindRoot}\left[\frac{1}{d}\,2^{-1-\frac{d}{2}+\frac{2}{p}}\,d^{\frac{d\,(-2+p)}{2\,p}}\,p^{d/p}\,\pi^{-\frac{1}{2}+\frac{1}{p}}\,\text{Gamma}\left[\frac{1+d}{2}\right]^{\frac{-2+p}{p}}-d\,\frac{p-2}{2\,p}\ /.\ d\to 1,\ \{p,\ 2\}\right]$$

Out[ ]= {p → 7.47622}